

マイコン FM-77 による節理データ解析

Computer program of equal area projection

BASICプログラム・・・ 島 津 定 雄*

Sadao SHIMAZU

機械語サブルーチン・・・ 堀 川 春 樹*

Haruki HORIKAWA

(1) 概 要

•最大で220の節理データをシュミット投影法により解析し、ポイントダイアグラムとコンターダイアグラムを描く。

•ディスプレイ画面上のコンターダイアグラムから任意の点を指定して、その点を極とする面の走向・傾斜を求める。

•プログラムの一部を機械語にすることにより計算処理を高速化した。

•プリンターに出力されるものは、以下の通りである。


- ① 入力した節理データの一覧表 (第1表)。
- ② シュミットネット上に節理面の極をプロットした図 (第1図)。
- ③ シュミット法により処理した頻度データの一覧表 (第2表)。
- ④ コンターダイアグラム (第2図)。
- ⑤ コンターダイアグラム上の任意の点を極とする面の走向・傾斜。

第1表 入 力 デ ー タ

JOINT DATA			file name = ARAKI								
No. 1	N45E	21S	No. 2	N36W	86N	No. 3	N85W	82S	No. 4	N69W	69N
No. 5	N72W	79N	No. 6	N66W	85N	No. 7	N68W	80N	No. 8	N67W	88S
No. 9	N72W	82S	No. 10	N43E	69S	No. 11	N57W	77S	No. 12	N34E	69S
No. 13	N72W	86N	No. 14	N68W	83S	No. 15	N42E	V	No. 16	N60W	84S
No. 17	N89W	87S	No. 18	N87W	74N	No. 19	N44W	77N	No. 20	N23E	78S
No. 21	N58W	81S	No. 22	N17E	44S	No. 23	N62W	77S	No. 24	N27E	52S
No. 25	N73E	86S	No. 26	N85W	86S	No. 27	N28W	69N	No. 28	N54W	72N
No. 29	N64W	79N	No. 30	N79W	63N	No. 31	N42W	65N	No. 32	N59W	74N
No. 33	N74W	80S	No. 34	N63W	74S	No. 35	N33E	78S	No. 36	N35W	64N
No. 37	N43W	65N	No. 38	N44W	74N	No. 39	N58W	V	No. 40	N62W	78S
No. 41	N20W	44N	No. 42	N55W	80N	No. 43	N77W	78N	No. 44	N14E	68S
No. 45	N6E	76S	No. 46	N23W	57N	No. 47	N42W	72N	No. 48	N26W	78N
No. 49	N74E	49N	No. 49	N12E	V	No. 51	N25E	78N	No. 52	N8E	74N
No. 53	N18W	27S	No. 54	N52W	27S	No. 55	N38W	58N	No. 56	N6E	18S
No. 57	N83W	80N	No. 58	N54W	56N	No. 59	N73W	68N	No. 60	N84W	71N
No. 61	N86W	70N	No. 62	N69W	75N	No. 63	N70W	80E	No. 64	N86E	63N
No. 65	N76W	80N	No. 66	N87W	70N	No. 67	N70W	47S	No. 68	N4E	80E
No. 69	N61W	66S	No. 70	N33E	36S	No. 71	N4W	87N	No. 72	N72W	V
No. 73	NS	77W	No. 74	N56W	78S	No. 75	N16E	13S	No. 76	N36W	49N
No. 77	N5	66N	No. 78	N68W	83N	No. 79	N86E	80S	No. 80	N68W	V
No. 81	N47W	50S	No. 82	N63W	86N	No. 83	N68E	84N	No. 84	N25E	42N
No. 85	N7W	55N	No. 86	N14E	64S	No. 87	N36W	70E	No. 88	N10W	26S
No. 89	N14E	78E	No. 90	N65W	V	No. 91	N18E	64S	No. 92	N18W	60S
No. 93	N37W	56N	No. 94	N29E	75S	No. 95	N16W	67S	No. 96	N58W	V
No. 97	N74W	89S	No. 98	N85E	74S	No. 99	N76W	70S	No. 100	N58W	V
No. 101	N21E	53S	No. 102	N17W	68S	No. 103	N57W	84N	No. 104	N32W	82N
No. 105	N55W	81N	No. 106	N34E	73S	No. 107	N82W	83S	No. 108	N35W	67S
No. 109	N24E	12N	No. 110	N78W	80S	No. 111	N54W	82N	No. 112	N36E	53S
No. 113	N44W	25S	No. 114	N35E	38N	No. 115	N64W	80N	No. 116	N28E	28S
No. 117	N62E	28N	No. 118	N13W	44E	No. 119	N34E	50S	No. 120	N31W	77N

* 島根大学理学部地質学教室

JOINT DATA PLOT
 file name = ARAKI
 declination 7.0 w
 DATAS. 120




第1図 ポイントフィギュア

第2表 頻度 (%) 分布表

JOINT DATA		count %																				
		file name = ARAKI																				
		0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20
0	0	0	0	0	0	0	0	1	0	1	2	2	2	8	5	7	0	0	0	0	0	0
1	0	0	0	0	0	1	1	0	0	0	1	3	8	8	5	3	0	0	0	0	0	0
2	0	0	0	0	0	1	1	0	0	3	4	4	5	3	3	2	0	0	0	0	0	0
3	0	0	0	0	0	0	0	0	1	3	4	3	3	2	3	2	1	1	0	0	0	0
4	0	1	1	0	0	0	0	0	1	1	1	1	0	1	3	3	3	3	0	0	0	0
5	0	0	1	0	0	0	0	0	1	0	0	0	1	1	3	5	3	3	1	0	0	0
6	0	0	0	0	0	0	0	0	1	0	0	0	0	1	3	3	1	2	0	0	0	0
7	0	1	1	0	0	0	0	0	1	1	0	0	0	1	2	1	1	0	0	0	0	0
8	0	0	1	0	0	0	2	1	1	1	0	1	0	1	2	2	1	0	0	2	2	0
9	1	0	1	0	0	0	1	1	0	1	0	1	0	0	1	2	1	0	1	2	0	0
10	1	1	1	0	0	0	0	0	0	1	0	2	2	0	1	1	0	2	1	1	1	1
11	1	1	1	0	0	0	0	2	2	0	0	2	2	2	2	3	3	3	2	2	2	1
12	2	0	0	2	2	0	0	0	3	2	0	1	1	1	2	3	3	1	1	1	0	0
13	2	0	0	3	3	1	0	0	1	0	0	0	0	0	2	2	1	3	3	0	0	0
14	0	0	0	0	0	0	1	1	0	1	0	0	0	0	0	0	2	3	1	0	1	0
15	0	0	0	0	0	0	0	1	1	1	0	0	0	0	0	0	1	0	1	0	0	0
16	0	2	0	0	0	0	0	1	1	0	0	0	0	0	0	0	0	0	1	1	0	0
17	0	0	2	1	0	0	2	4	2	1	0	0	1	0	0	0	0	0	0	0	0	0
18	0	0	0	1	0	0	4	6	3	3	0	2	2	0	1	0	0	0	0	0	0	0
19	0	0	0	0	1	5	6	4	6	4	3	3	1	1	1	1	1	0	0	0	0	0
20	0	0	0	0	0	0	7	8	7	2	2	2	0	0	0	0	0	0	0	0	0	0

数値は、小数第1位を四捨五入してある。



第2図 コンターダイヤグラム

等頻度線は7本に設定してあり、各等頻度線のパレットコードが示す値は、図の右端に表示している。

(II) 使用機種

パソコン：富士通 FM-77

言語：F-BASIC V 3.0 (DISK モード)

プリンター：MB27407 または、ハードコピーにより真円を得る場合の楕円率が0.5のプリンター

ディスプレイ：カラー2000文字対応

(III) プログラムの入力

本プログラムは、計算の高速化のため、一部に機械語を使用している。入力するプログラムは、リスト1の BASIC プログラムと、リスト2の機械語プログラムである。機械語プログラムのアドレスは、下記の通りである。

START = &H 6 C 0 0

END = &H 6 F 2 D

EXEC = &H 6 C 0 0

ただし、機械語プログラムのファイル名は、「JOINT. M」にすること。

(IV) 使用法

BASIC プログラムを実行すると、まず機械語プログラムをロードする。次に、節理データの入力方法を選択する。すると、節理データ処理メニューになり、キー入力待ちになる。メニューの各機能は、以下の通りである。

1. 節理データの入力（最初にロードを選択した場合は、DATA LOAD となる）。
2. 節理データの修正。
3. 節理データの表示，印字。
4. 節理データの処理計算スタート。
5. 節理データの保存。
6. プログラムの再スタート，終了。

節理データを新規入力する場合，まずデータの総数と偏角補正値を，それぞれ入力する。節理データの入力形式は，下記の基準以外のものは受け入れないので注意すること。

走向データ（strike）

- 左の1字は「N」または「S」に限る。
- 数字（角度）は，90以下の自然数。
- 右の1字は「E」または「W」に限る。
- 例外として，EW，WE，NS，SNも認められる。

傾斜データ（dip）

- 数字（角度）は，90以下の自然数。
- 右の1字は，「N」，「S」，「E」，「W」に限る。
- 垂直な場合は「V」である。

節理データの入力例を第1表に示す。

次に，メニューの4番を選択すると，シュミットネット投影によるコンターダイアグラム解析が始まる。まず第1図のポイントフィギュアのハードコピーをとるかどうかが選択する。すると，節理面の極が画面上にプロットされ，続いてシュミット法による頻度分布計算処理が始まる。この計算は，機械語サブルーチンが担当しており，計算時間は節理データ数が120の場合で15分（BASIAだけのプログラムでは約70分），データ数が200の場合で25分程である。計算は，浮動小数点の実数計算で行われる。

計算が終わると，頻度データ処理メニュー画面となり，キー入力待ちになる。メニューの各機能は，以下の通りである。

1. 頻度分布表を画面に表示する。
2. 頻度分布表（第2表）を印字する。
3. 頻度分布を，カラーパターン化して画面に表示する。
4. コンターダイアグラム作画。
5. 節理データ処理メニューに戻る。

ここで，メニューの4を選択すると，コンターダイアグラムが得られる。作画に要する時間は10分程である。そのあと，グラフィックカーソルを使って，コンターダイアグラムから結果を読みとる作業を行なう。

白い十字型のグラフィックカーソルが画面に現れたら，カーソルキーを使って任意の位置に移動させ，リターンキーを押す（グラフィックカーソルについては，FM-77のBASIC文法書を参照）。すると，その点を極とする面の走向・傾斜が画面に表示される。コンターダイアグラムをハードコ

ピーしている場合にはプリンターにこの結果が出力される。

このあと、操作を続ける場合は、画面左下に表示されるメッセージに従い、キー入力する。

参考文献

地学団体研究会 (1983) 堆積物の研究法—礫岩・砂岩・泥岩—, 地学双書24.

酒井 謙二 (1984) 等高線作画プログラム. I/O 8月号, 工学社.

RAGAN, D. A. (1985) Structural Geology—An introduction to geometrical technique, Wiley.

〔リスト1〕

```

10 : save"1:JOINT" ' Jun 30 '86
20 :
30 :
40 : BASIC Program by S.Shimazu
50 : machine Program by H.Horikawa
60 :
70 :
100 WIDTH 40,20:CLS:COLOR 4:CLEAR 3500,&H6BF:DEFINT I-N,S
110 DIM SN(220),S$(220),D$(220),AX(220),AY(220),B(20,20)
120 PRINT " - object LOAD -"
130 PRINT:COLOR 5:PRINT "Set the DISK on Drive 1:":GOSUB 1570
140 IF N$="N" THEN 120 ELSE LOADM"1:JOINT.M"
150 PRINT "----- JOINT DATA CALCULATION -----":PRINT
160 PRINT "DATA INPUT ... Key OR LOAD?":PRINT:CC=0:CP=0
170 M0$=" 1 ..... DATA INPUT"
180 M1$=" 1 ..... DATA LOAD"
190 PRINT SPC(10):"( LOAD:SPC / Key:else )"
200 W$=INKEY$:IF W$="" THEN 200
210 IF W$="" THEN L$(1)=M1$ ELSE L$(1)=M0$
220 L$(2)=" 2 ..... DATA REPAIR"
230 L$(3)=" 3 ..... DATA INDICATE"
240 L$(4)=" 4 ..... CALCULATION"
250 L$(5)=" 5 ..... DATA SAVE"
260 L$(6)=" 6 ..... START / END"
270 WIDTH 40,25:CLS:"ON ERROR GOTO 240
280 IF CC=1 THEN CP=0:RP=0 ELSE CP=1:RP=1
290 PRINT "----- JOINT DATA MENU -----"
300 PRINT SPC(30):"DATAS.":COLOR 7:PRINT N0:COLOR 4
310 IF N0=0 THEN PRINT L$(1)
320 PRINT L$(2),L$(3),L$(4),L$(5),L$(6)
330 PRINT:INPUT " ACTION No.",W0
340 IF W0>6 OR W0<1 THEN 330
350 FOR I=1 TO 6:IF W0=I THEN PRINT:COLOR 6:PRINT L$(I)
360 NEXT I:GOSUB 1570:IF N$="N" THEN GOTO 270
370 ON W0 GOSUB 390,520,680,900,1800,860
380 CLS:COLOR 4:GOTO 270
390 IF N0>0 THEN 270
400 IF L$(1)<>M0$ THEN 1630
410 ' data input
420 INPUT "How Many DATAS (max,220) ? ",N0:N=N0-1:IF N0>220 THEN 420
430 PRINT:INPUT "declination angle ° = West ",WP
440 GOSUB 1570:IF N$="N" THEN 420
450 FOR I=0 TO N:SN(I)=I+1
460 PRINT "DATA No.":S$(I):COLOR 7:PRINT SN(I):COLOR 4:INPUT "strike ? = ",S$(I)
470 PRINT TAB(27):":":INPUT "dip ? = ",D$(I)
480 GOSUB 1570:IF N$="N" THEN BEEP:GOTO 460
490 NEXT I
500 RETURN
510 ' data repair
520 INPUT "No. OF joint ? ",W1
530 FOR I=0 TO N
540 IF SN(I)=W1 THEN 570
550 NEXT I:PRINT
560 PRINT "UNFIND joint No.":W1:FOR I=1 TO 2000:NEXT:GOTO 650
570 PRINT:PRINT "..... joint No.":SN(I)
580 PRINT "strike =":S$(I):PRINT " dip =":D$(I)
590 GOSUB 1570:IF N$="N" THEN RETURN
600 INPUT "new strike = ",A0$:IF A0$="" THEN A0$=S$(I)
610 INPUT "new dip = ",B0$:IF B0$="" THEN B0$=D$(I)
620 S$(I)=A0$:D$(I)=B0$
630 PRINT:PRINT " - back to the MENU -"
640 GOSUB 1570:IF N$="N" THEN 510 ELSE 650
650 IF CC=1 THEN FOR I=0 TO 20:FOR J=0 TO 20:B(I,J)=0:NEXT J,I
660 CC=0:RP=1:RETURN
670 ' joint data indicate
680 PRINT "JOINT DATA INDICATE TABLE":PRINT
690 PRINT " ( PRINT OUT:SPC CRT:else )":
700 W$=INKEY$:IF W$="" THEN 700
710 IF W$="" THEN COLOR 7:PRINT "printer":ELSE PRINT "CRT"
720 GOSUB 1570:IF N$="N" THEN 840
730 IF W$="" THEN 790 ELSE 740

```

```

740 CLS:PRINT"file name = ";GT$:PRINT:FOR I=0 TO N
750 PRINT"DATA No. ";SN(I);
760 PRINT TAB(14);"strike = ";S$(I);
770 PRINT TAB(27);"dip = ";D$(I);
780 NEXT I:PRINT:GOSUB 1530:GOTO 840
790 LPRINT:LPRINT TAB(22);"JOINT DATA      file name = ";GT$
800 LPRINT:FOR I=0 TO N STEP 2
810 LPRINT TAB(5);"No. ";SN(I);LPRINT TAB(14);"strike = ";S$(I);LPRINT TAB(30);
      "dip = ";D$(I);
820 LPRINT TAB(42);"I No. ";SN(I+1);LPRINT TAB(53);"strike = ";S$(I+1);LPRINT T
      AB(69);"dip = ";D$(I+1)
830 NEXT I
840 RETURN
850 '-----end/start-----
860 COLOR 1:PRINT"... NEXT OPERATION START ? ... "
870 GOSUB 1570:IF N$="N" THEN 880:ELSE RUN 150
880 COLOR 3:PRINT"... THIS PROGRAM END ...":GOSUB 1570
890 IF N$="N" THEN GOTO 380 ELSE WIDTH 80,25:END
900 '-----stereo net-----
910 PRINT"JOINT DATA PLOT by Schmidt's net":PRINT
920 PRINT "( HARD COPY:SPC CRT:else ):";
930 W$=INKEY$:IF W$=" " THEN 930
940 IF W$=" " THEN U1= 5:COLOR 7:PRINT"HARD COPY":ELSE U1=.4495:PRINT"CRT"
950 GOSUB 1570:IF N$="N" THEN 920
960 GOTO 990
970 PRINT:IF U1=.5 THEN LPRINT:HARD2:GOSUB 2060 ELSE GOSUB 2060:GOSUB 1530
980 CC=1:GOSUB 2390:RETURN
990 X=332:R0=180:IF U1=.5 THEN Y=104 ELSE Y=97
1000 E=100/N0:COLOR 4:WIDTH 80,25
1010 CLS:CIRCLE(X,Y),R0,7,U1:GOSUB 1430
1020 FOR I=0 TO N
1030 '-----data set-----
1040 IF LEN(S$(I))=3 THEN FS=VAL(MID$(S$(I),2,1))
1050 IF LEN(S$(I))=4 THEN FS=VAL(MID$(S$(I),2,2))
1060 IF LEN(D$(I))=1 THEN FD=0
1070 IF LEN(D$(I))=2 THEN FD=VAL(LEFT$(D$(I),1))
1080 IF LEN(D$(I))=3 THEN FD=VAL(LEFT$(D$(I),2))
1090 L$=LEFT$(S$(I),1):R$=RIGHT$(S$(I),1)
1100 IF L$="S" AND R$="E" THEN R$="W":L$="N"
1110 IF L$="S" AND R$="W" THEN R$="E":L$="N"
1120 B$=RIGHT$(D$(I),1):IF B$="V" THEN FD=90:B$="S"
1130 IF R$="E" AND B$="E" THEN B$="S"
1140 IF R$="E" AND B$="W" THEN B$="N"
1150 IF R$="W" AND B$="E" THEN B$="N"
1160 IF R$="W" AND B$="W" THEN B$="S"
1170 IF S$(I)="WE" OR S$(I)="EW" THEN FS=90
1180 IF S$(I)="NS" OR S$(I)="SN" THEN FS=0
1190 '-----joint plot cul-----
1200 Z=3.14159/180:IF R$="W" THEN FS=180-FS
1210 FS=FS-WP
1220 IF FS<0 AND B$="S" THEN B$="N":FS=180+FS
1230 IF FS<0 AND B$="N" THEN B$="S":FS=180+FS
1240 X0=SIN(Z*FS)*R0
1250 Y0=COS(Z*FS)*R0
1260 L=SIN(Z*FD)*R0:D0=COS(Z*FD)*R0
1270 D=SQR(D0^2+(R0-L)^2)/SQR(2):D=R0-D
1280 IF S$(I)="NS" OR S$(I)="SN" THEN 1310
1290 IF B$="N" THEN 1360
1300 IF B$="S" THEN 1340
1310 '-----plot ns-----
1320 IF B$="E" THEN 1340
1330 IF B$="W" OR B$="S" THEN 1380
1340 '-----plot s-----
1350 X2=X+Y0*D/R0:Y1=Y+X0*D/R0*U1:Y2=90+X0*D/R0
1360 PSET(X2,Y1,4):IF U1=.5 THEN PSET(X2+1,Y1,4)
1370 AX(I)=X2:AY(I)=Y2+90:NEXT I:GOTO 970
1380 '-----plot n-----
1390 X2=X-Y0*D/R0:Y1=Y-X0*D/R0*U1:Y2=90-X0*D/R0
1400 PSET(X2,Y1,4):IF U1=.5 THEN PSET(X2+1,Y1,4)
1410 AX(I)=X2:AY(I)=Y2+90:NEXT I:GOTO 970
1420 '-----stereo net information-----
1430 LINE(X,Y-R0*U1-5)-(X,Y-R0*U1),PSET,7
1440 LINE(X,Y+R0*U1+5)-(X,Y+R0*U1),PSET,7
1450 LINE(X-R0-5/U1,Y)-(X-R0,Y),PSET,7
1460 LINE(X+R0+5/U1,Y)-(X+R0,Y),PSET,7
1470 COLOR 7:LOCATE 41,0:PRINT"N"
1480 PRINT"JOINT DATA PLOT":COLOR 4
1490 PRINT:PRINT"file name = ";GT$
1500 PRINT:PRINTUSING"declination ##.# w":WP
1510 PRINT:PRINT"DATAS. ";N0:PRINT
1520 RETURN
1530 '-----Hit SPC Key-----
1540 COLOR 5:PRINT"... Hit SPC Key ...":BEEP:COLOR 4
1550 Y$=INKEY$:IF Y$=" " THEN 1550
1560 IF Y$=" " THEN RETURN ELSE 1550
1570 '-----Are You Sure?-----
1580 COLOR 2:PRINT:PRINT"Are You Sure ? ( Y:else N:n )"
1590 N$=INKEY$:IF N$=" " THEN 1590
1600 IF N$="n" OR N$="N" THEN N$="N"
1610 COLOR 4:PRINT:RETURN
1620 '-----joint data load-----

```

```

1630 PRINT "FILES INDICATE ? ( Y:SPC N:else ):";
1640 W$=INKEY$:IF W$=" " THEN 1640
1650 IF W$<" " THEN PRINT "N":GOTO 1680:ELSE PRINT "Y"
1660 COLOR 5:PRINT:PRINT "Set the DISK on Drive 1:":GOSUB 1570
1670 IF N$="N" THEN 1630 ELSE FILES=1:
1680 PRINT:INPUT "file name ? " :GT$:IF LEN(GT$)>8 THEN 1680
1690 PRINT:COLOR 5:PRINT "Set the DISK on Drive 1:":GOSUB 1570
1700 IF N$="N" THEN RETURN
1710 CLS:PRINT "----- DATA LOADING -----"
1720 OPEN "I",#1,"1:"+GT$:INPUT #1,WP
1730 I=0
1740 INPUT #1,SN(I),S$(I),D$(I)
1750 IF EOF(1) THEN GOTO 1770
1760 I=I+1:N=I:GOTO 1740
1770 CLOSE #1:N0=N+1
1780 BEEP:RETURN
1790 '----- joint data save -----
1800 PRINT:INPUT "file name ? " :GT$:IF LEN(GT$)>8 THEN 1800
1810 PRINT:COLOR 5:PRINT "Set the DISK on Drive 1:":GOSUB 1570
1820 IF N$="N" THEN GOTO 1890
1830 CLS:PRINT "----- DATA SAVING -----"
1840 OPEN "O",#1,"1:"+GT$:PRINT #1,WP
1850 FOR I=0 TO N
1860 PRINT #1,SN(I),S$(I);",":D$(I)
1870 NEXT I
1880 CLOSE #1
1890 RETURN
1900 '----- count data print out -----
1910 LPRINT:LPRINT "JOINT DATA count % file name = ":GT$:LPRINT
1920 LPRINT " ":FOR I=0 TO 20:LPRINTUSING "##";I:NEXT I
1930 LPRINT:LPRINT
1940 FOR I=0 TO 20:LPRINTUSING "## ":I:FOR J=0 TO 20
1950 IF J=20 THEN LPRINTUSING "###";B(J,I):ELSE LPRINTUSING "###";B(J,I);
1960 NEXT J,I
1970 RETURN
1980 '----- count data indicate -----
1990 CLS:PRINT "JOINT DATA count % file name = ":GT$:PRINT
2000 PRINT " ":FOR I=0 TO 20:PRINTUSING "##";I:NEXT I
2010 PRINT:PRINT
2020 FOR I=0 TO 20:PRINTUSING "## ":I:FOR J=0 TO 20
2030 IF J=20 THEN PRINTUSING "###";B(J,I):ELSE PRINTUSING "###";B(J,I);
2040 NEXT J,I
2050 LOCATE 65,23:GOSUB 1530:RETURN
2060 '----- data count -----
2070 IF CC<>0 THEN RETURN
2080 O=&H7040:P=0:P1=0:P2=0:P3=0
2090 P=VARPTR(X):POKE O+&H0,P#256:POKE O+&H1,P MOD256
2100 P=VARPTR(Y):POKE O+&H2,P#256:POKE O+&H3,P MOD256
2110 P=VARPTR(R0):POKE O+&H4,P#256:POKE O+&H5,P MOD256
2120 P=VARPTR(U1):POKE O+&H6,P#256:POKE O+&H7,P MOD256
2130 P=VARPTR(AX(0)):POKE O+&H8,P#256:POKE O+&H9,P MOD256
2140 P=VARPTR(AY(0)):POKE O+&HA,P#256:POKE O+&HB,P MOD256
2150 P=VARPTR(E):POKE O+&HC,P#256:POKE O+&HD,P MOD256
2160 P=VARPTR(B(0,0)):POKE O+&HE,P#256:POKE O+&HF,P MOD256
2170 POKE &H7050,N
2180 O=&H7030
2190 P1=1.1:P=VARPTR(P1):POKE O+&H0,P#256:POKE O+&H1,P MOD256
2200 P2=.9:P=VARPTR(P2):POKE O+&H2,P#256:POKE O+&H3,P MOD256
2210 P3=17.8:P=VARPTR(P3):POKE O+&H4,P#256:POKE O+&H5,P MOD256
2220 BEEP:TIME$="00:00:00":EXEC &H6C00:PRINT TIME$
2230 PRINT:RETURN
2240 '----- pattern -----
2250 GOSUB 2560:CLS:WIDTH 80,25
2260 LINE(103,15)-(441,184)PSET,7,B
2270 FOR I=0 TO 20:FOR J=0 TO 20
2280 CV=INT(B(J,I)/DELH):IF CV>7 THEN CV=7
2290 LOCATE J*2+13,I+2:COLOR CV:PRINT "■";
2300 NEXT J,I
2310 CIRCLE(272,100),161.7,5
2320 LINE(103,15)-(441,184)PSET,8,B:COLOR 7
2330 LINE(272,20)-(272,10)PSET,7:LINE(272,188)-(272,180)PSET,7
2340 LINE(110,100)-(96,100)PSET,7:LINE(433,100)-(449,100)PSET,7
2350 FOR K=1 TO 7:LOCATE 60,K*2+1:COLOR 8-K:PRINT "■":COLOR 4:PRINT " ":8-K
2360 NEXT K:COLOR 7:LOCATE 56,1:PRINT "PLOT DATA pattern"
2370 SYMBOL(268,0) "N",1,15
2380 LOCATE 57,17:GOSUB 1530:RETURN
2390 '----- count data graphics -----
2400 CLS:WIDTH 40,20:COLOR 4
2410 R$(1)=" 1. count data indicate"
2420 R$(2)=" 2. count data print out"
2430 R$(3)=" 3. pattern display"
2440 R$(4)=" 4. contour diagram draw"
2450 R$(5)=" 5. back to the menu"
2460 PRINT "----- COUNT DATA GRAPHICS -----"
2470 PRINT:FOR I=1 TO 5:PRINT R$(I):NEXT I
2480 PRINT:INPUT " ACTION No.",W1
2490 IF W1<1 OR W1>5 THEN 2480
2500 FOR I=1 TO 5:IF W1=I THEN COLOR 6:PRINT:PRINT R$(I)
2510 NEXT I:IF W1=2 THEN COLOR 7:PRINT:PRINT "printer on line"
2520 GOSUB 1570:IF N$="N" THEN 2400
2530 WIDTH 80,25:IF W1=5 THEN RETURN

```

```

2540 ON W1 GOSUB 1980,1900,2240,2630
2550 GOTO 2400
2560 IF CP=0 AND RP=0 THEN RETURN ELSE PRINT". wait a moment ."
2570 HMIN=B(0,0):HMAX=B(0,0):RP=0:CP=0
2580 FOR I=0 TO 20:FOR J=0 TO 20
2590 IF HMIN>B(I,J) THEN HMIN=B(I,J)
2600 IF HMAX<B(I,J) THEN HMAX=B(I,J)
2610 NEXT J,I:DELH=(HMAX-HMIN)/7:RETURN
2620 _____ contour line draw _____
2630 WIDTH 40,20:PRINT"contour diagram Draw":PRINT
2640 PRINT" ( HARD COPY:SPC CRT:else ):";
2650 W$=INKEY$:IF W$="" THEN 2650
2660 IF W$=" " THEN COLOR 7:PRINT"HARD COPY":U2=.5:ELSE PRINT"CRT":U2=.4495
2670 GOSUB 1570:IF N$="N" THEN RETURN
2680 GOSUB 2560:NK=7:NK1=8:FOR I=1 TO NK1
2690 T(I)=HMAX-DELH*(I-1):NEXT I
2700
2710 CLS:WIDTH 80,25:NK1=19:NY1=19
2720 FOR K=1 TO NK:NCOL=8-K:ZM=T(K):IF K<>1 THEN ZM=ZM+.0001
2730 FOR J=0 TO NY1:J1=J+1
2740 FOR I=0 TO NX1:I1=I+1:M=0
2750 XK(1)=0:XK(2)=0:XK(3)=0:XK(4)=0
2760 YK(1)=0:YK(2)=0:YK(3)=0:YK(4)=0
2770 H1=B(J,I):H2=B(J,I1):H3=B(J1,I1):H4=B(J1,I)
2780 H12=(H1-ZM)*(H2-ZM)
2790 H23=(H2-ZM)*(H3-ZM)
2800 H34=(H3-ZM)*(H4-ZM)
2810 H41=(H4-ZM)*(H1-ZM)
2820 IF H12>0 AND H23>0 AND H34>0 THEN 3210
2830 X1=J*18+152:X4=J1*18+152:X3=X4:X2=X1
2840 Y4=I1*18*U2+.5:Y1=Y4:Y2=(I+2)*18*U2+.5:Y3=Y2
2850 IF H12<=0 THEN 2830
2860 IF H23<=0 THEN 2970
2870 IF H34<=0 THEN 3040
2880 GOTO 3210
2890 IF H1=H2 THEN GOTO 3170
2900 M=M+1:ZK=(ZM-H1)/(H2-H1)
2910 XK(M)=ZK*(X2-X1)+X1
2920 YK(M)=ZK*(Y2-Y1)+Y1
2930 IF H23<=0 THEN 2970
2940 IF H34<=0 THEN 3040
2950 IF H41<=0 THEN 3100
2960 GOTO 3140
2970 IF H2=H3 THEN GOTO 3180
2980 M=M+1:ZK=(ZM-H2)/(H3-H2)
2990 XK(M)=ZK*(X3-X2)+X2
3000 YK(M)=ZK*(Y3-Y2)+Y2
3010 IF H34<=0 THEN 3040
3020 IF H41<=0 THEN 3100
3030 GOTO 3140
3040 IF H3=H4 THEN GOTO 3190
3050 M=M+1:ZK=(ZM-H3)/(H4-H3)
3060 XK(M)=ZK*(X4-X3)+X3
3070 YK(M)=ZK*(Y4-Y3)+Y3
3080 IF H41<=0 THEN 3100
3090 GOTO 3140
3100 IF H4=H1 THEN GOTO 3200
3110 M=M+1:ZK=(ZM-H4)/(H1-H4)
3120 XK(M)=ZK*(X1-X4)+X4
3130 YK(M)=ZK*(Y1-Y4)+Y4
3140 LINE(XK(1),YK(1))-(XK(2),YK(2)),PSET,NCOL
3150 LINE(XK(3),YK(3))-(XK(4),YK(4)),PSET,NCOL
3160 GOTO 3210
3170 LINE(X1,Y1)-(X2,Y2),PSET,NCOL:GOTO 3210
3180 LINE(X2,Y2)-(X3,Y3),PSET,NCOL:GOTO 3210
3190 LINE(X3,Y3)-(X4,Y4),PSET,NCOL:GOTO 3210
3200 LINE(X4,Y4)-(X1,Y1),PSET,NCOL
3210 NEXT I,J,K
3220 _____ contour information _____
3230 LINE(0,0)-(639,199),PSET,7,B
3240 CY=360*U2/2+(18*U2+.5):CX=332:R2=182
3250 CIRCLE(CX,CY),R2,7,U2
3260 PAINT(162,20),6,7:PAINT(502,20),6,7
3270 PAINT(162,375*U2),6,7:PAINT(502,375*U2),6,7
3280 PAINT(162,20),8,7:PAINT(502,20),8,7
3290 PAINT(162,375*U2),8,7:PAINT(502,375*U2),8,7
3300 LINE(0,0)-(639,199),PSET,8,B
3310 LINE(CX,18*U2+.4)-(CX,8),PSET,7
3320 LINE(CX,378*U2+.6)-(CX,378*U2+10),PSET,7
3330 LINE(CX-R2,CY)-(CX-R2-8,CY),PSET,7
3340 LINE(CX+R2+8,CY)-(CX+R2,CY),PSET,7
3350 COLOR 7:LOCATE 41,0:PRINT"N"
3360 LOCATE 0,1:PRINT"Contour diagram"
3370 COLOR 4:PRINT:PRINT" file name = ";GT$
3380 PRINT:PRINT"DATAS.":N0
3390 PRINT:PRINT"contour step."
3400 PRINT:PRINTUSING"###.### %";DELH
3410 LOCATE 66,4:PRINT"contour lines."
3420 FOR I=1 TO NK:LOCATE 69,5+I*2:COLOR 8-I:PRINT 8-I:COLOR 4
3430 PRINTUSING"###.### %":T(I):NEXT I
3440 IF U2=.5 THEN LPRINT:HARDC2

```



```

3450 LOCATE 0,16:GOSUB 1530
3460 '----- point pick up -----
3470 Z1=180/3.14159:LOCATE 0,16:PRINT " POINT SET "
3480 GCURSOR(CX,CY),(XG,YG),7
3490 XG0=XG-CX:YG0=(YG-CY)/U2:IF SQR(XG0^2+YG0^2)>R0*1.02 THEN 3480
3500 IF YG0=0 THEN AG=90 ELSE AG=ATN(XG0/YG0)*Z1
3510 AW=SGN(AG):AS0=INT(AG+.5):IF AW=1 THEN AS$="E":AS=90-AG
3520 IF AW=-1 THEN AS$="W":AS=90+AG
3530 AAS$=STR$(INT(AS+.5)):S1$="N"+AAS$+" "+AS$
3540 IF AS0=0 THEN S1$="EW"
3550 IF VAL(AAS$)=0 THEN S1$="NS"
3560 IF AS0=0 THEN S1$="EW"
3570 BG=SQR(XG0^2+YG0^2):BG1=BG*SQR(2)
3580 D1=BG*SQR(2)/2:IF D1=0 THEN D1=.00001
3590 L1=SQR(R0^2-D1^2):B1=ATN(L1/D1)*Z1:B1=90-B1:B2=B1*2
3600 IF YG-CY>0 THEN AD$="N"
3610 IF YG-CY<0 THEN AD$="S"
3620 IF S1$="NS" AND XG0>0 THEN AD$="W"
3630 IF S1$="NS" AND XG0<0 THEN AD$="E"
3640 AAD$=STR$(INT(B2+.5)):D1$=AAD$+" "+AD$
3650 IF INT(B2+.5)=90 THEN D1$="V"
3660 IF INT(B2+.5)=0 THEN D1$="H"
3670 BEEP:PSET(XG,YG,7)
3680 '----- pick up point plane -----
3690 E1$=" continue:y ":E2$="back to menu:N":CB=1
3700 E$=E1$+E2$:GOSUB 3790:LOCATE 0,23:COLOR 7
3710 W$=INKEY$:IF W$="" THEN 3710
3720 IF W$="Y" THEN PRINT E1$:GOTO 3740
3730 IF W$="N" THEN PRINT SPC(14):E2$:ELSE 3710
3740 FOR I=0 TO 2000:NEXT I:COLOR 2:BEEP
3750 LOCATE 0,23:PRINT "You Sure ? ( Y:y N:else ) ":COLOR 4
3760 Y$=INKEY$:IF Y$="" THEN 3760 ELSE IF Y$<>"y" THEN CB=0:GOTO 3700
3770 IF W$="N" THEN RETURN
3780 IF W$="Y" THEN E$="":S1$="":D1$="":CB=0:GOSUB 3790:GOTO 3480
3790 LOCATE 0,18:PRINTUSING"strike = & . &";S1$
3800 LOCATE 0,20:PRINTUSING" dip = & &";D1$
3810 LOCATE 0,23:PRINTUSING"& &";E$
3820 IF U2=.5 AND CB=1 THEN LPRINT:LPRINTUSING" - POLE - &";E$ strike = &
& dip = & &";S1$;D1$
3830 CB=1:RETURN
3840 '
3850 ' file name = JOINT
3860 '

```

